

pkcross — Analyze crossover experiments

[Description](#)
[Options](#)
[Also see](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Methods and formulas](#)

[Syntax](#)
[References](#)

Description

`pkcross` analyzes data from a crossover design experiment. When analyzing pharmaceutical trial data, if the treatment, carryover, and sequence variables are known, the omnibus test for separability of the treatment and carryover effects is calculated.

`pkcross` is one of the `pk` commands. Please read [\[R\] pk](#) before reading this entry.

Quick start

For pharmacokinetic outcome `y` measured for subjects identified by `idvar` given treatments `treat` in sequences identified by `seq` in periods `period` with potential carryover effects from previous treatment `carry`:

Sequence, treatment, and period effects for a 2×2 design

```
pkcross y, param(3) id(idvar) sequence(seq) treatment(treat) ///
      period(period)
```

Same as above, but estimate the carryover effect instead of the sequence effect

```
pkcross y, param(1) id(idvar) treatment(treat) period(period) ///
      carryover(carry)
```

Only estimate sequence, treatment, and period effects in higher-order designs

```
pkcross y, id(idvar) sequence(seq) treatment(treat) carryover(none)
```

Also estimate carryover effect and omnibus measure of separability of treatment and carryover effects

```
pkcross y, model(seq / idvar|seq treat carry period) ///
      treatment(treat) carryover(carry) sequence(seq)
```

Menu

Statistics > Epidemiology and related > Other > Analyze crossover experiments

Syntax

```
pkcross outcome [if] [in] [, options]
```

<i>options</i>	Description
Model	
<code>sequence(<i>varname</i>)</code>	sequence variable; default is <code>sequence(sequence)</code>
<code>treatment(<i>varname</i>)</code>	treatment variable; default is <code>treatment(treat)</code>
<code>period(<i>varname</i>)</code>	period variable; default is <code>period(period)</code>
<code>id(<i>varname</i>)</code>	ID variable; default is <code>id(id)</code>
<code>carryover(<i>varname</i>)</code>	carryover variable; default is <code>carryover(carry)</code>
<code>carryover(none)</code>	omit carryover effects from model; default is <code>carryover(carry)</code>
<code>model(<i>string</i>)</code>	specify the model to fit
<code>sequential</code>	estimate sequential instead of partial sums of squares
Parameterization	
<code>param(3)</code>	estimate mean and the period, treatment, and sequence effects; assume no carryover effects exist; the default
<code>param(1)</code>	estimate mean and the period, treatment, and carryover effects; assume no sequence effects exist
<code>param(2)</code>	estimate mean, period and treatment effects, and period-by-treatment interaction; assume no sequence or carryover effects exist
<code>param(4)</code>	estimate mean, sequence and treatment effects, and sequence-by-treatment interaction; assume no period or carryover effects exist

Options

Model

`sequence(varname)` specifies the variable that contains the sequence in which the treatment was administered. If this option is not specified, `sequence(sequence)` is assumed.

`treatment(varname)` specifies the variable that contains the treatment information. If this option is not specified, `treatment(treat)` is assumed.

`period(varname)` specifies the variable that contains the period information. If this option is not specified, `period(period)` is assumed.

`id(varname)` specifies the variable that contains the subject identifiers. If this option is not specified, `id(id)` is assumed.

`carryover(varname|none)` specifies the variable that contains the carryover information. If `carry(none)` is specified, the carryover effects are omitted from the model. If this option is not specified, `carryover(carry)` is assumed.

`model(string)` specifies the model to be fit. For higher-order crossover designs, this option can be useful if you want to fit a model other than the default. However, `anova` (see [R] [anova](#)) can also be used to fit a crossover model. The default model for higher-order crossover designs is outcome predicted by sequence, period, treatment, and carryover effects. By default, the model statement is `model(sequence period treat carry)`.

`sequential` specifies that sequential sums of squares be estimated.

Parameterization

`param(#)` specifies which of the four parameterizations to use for the analysis of a 2×2 crossover experiment. This option is ignored with higher-order crossover designs. The default is `param(3)`. See the [technical note](#) in this entry for 2×2 crossover designs for more details.

`param(3)` estimates the overall mean, the period effects, the treatment effects, and the sequence effects, assuming that no carryover effects exist. This is the default parameterization.

`param(1)` estimates the overall mean, the period effects, the treatment effects, and the carryover effects, assuming that no sequence effects exist.

`param(2)` estimates the overall mean, the period effects, the treatment effects, and the period-by-treatment interaction, assuming that no sequence or carryover effects exist.

`param(4)` estimates the overall mean, the sequence effects, the treatment effects, and the sequence-by-treatment interaction, assuming that no period or carryover effects exist. When the sequence-by-treatment interaction is equivalent to the period effect, this reduces to the third parameterization.

Remarks and examples

[stata.com](#)

`pkcross` is designed to analyze crossover experiments. Use `pkshape` first to reshape your data; see [\[R\] pkshape](#). `pkcross` assumes that the data were reshaped by `pkshape` or are organized in the same manner as produced with `pkshape`. Washout periods are indicated by the number 0. See the technical note below for more information on analyzing 2×2 crossover experiments.

□ Technical note

The 2×2 crossover design cannot be used to estimate more than four parameters because there are only four pieces of information (the four cell means) collected. `pkcross` uses ANOVA models to analyze the data, so one of the four parameters must be the overall mean of the model, leaving just 3 degrees of freedom to estimate the remaining effects (period, sequence, treatment, and carryover). Thus, the model is overparameterized. Estimation of treatment and carryover effects requires the assumption of either no period effects or no sequence effects. Some researchers maintain that estimating carryover effects at the expense of other effects is a bad idea. This is a limitation of this design. `pkcross` implements four parameterizations for this model. They are numbered sequentially from one to four and are described in [Options](#).

□

▷ Example 1

Consider the example data published in [Chow and Liu \(2009, 71\)](#). There are 24 patients, 12 in each sequence. Sequence 1 is the reference formulation followed by the test formulation; sequence 2 is the test formulation followed by the reference formulation. After reshaping the data with `pkshape`, we have variables that identify the subjects, periods, treatments, sequence, and carryover treatment. The `outcome` variable contains the reported $AUC_{0,t_{max}}$. To compute the ANOVA table, use `pkcross`:

```

. use https://www.stata-press.com/data/r18/chowliu
. pkshape id seq period1 period2, order(RT TR)
. pkcross outcome

Sequence variable = sequence
Period variable = period
Treatment variable = treat
Carryover variable = carry
ID variable = id

```

Analysis of variance (ANOVA) for a 2x2 crossover study

Source of variation	Partial SS	df	MS	F	Prob > F
Intersubjects					
Sequence effect	276.00	1	276.00	0.37	0.5468
Residuals	16211.49	22	736.89	4.41	0.0005
Intrasubjects					
Treatment effect	62.79	1	62.79	0.38	0.5463
Period effect	35.97	1	35.97	0.22	0.6474
Residuals	3679.43	22	167.25		
Total	20265.68	47			

Omnibus measure of separability of treatment and carryover = 29.2893%

There is evidence of intersubject variability, but there are no other significant effects. The omnibus test for separability is a measure reflecting the degree to which the study design allows the treatment effects to be estimated independently of the carryover effects. The measure of separability of the treatment and carryover effects indicates approximately 29% separability, which can be interpreted as the degree to which the treatment and carryover effects are orthogonal. This is a characteristic of the design of the study. For a complete discussion, see [Ratkowsky, Evans, and Alldredge \(1993\)](#). Compared with the output in [Chow and Liu \(2009\)](#), the sequence effect is mislabeled as a carryover effect. See [Ratkowsky, Evans, and Alldredge \(1993, sec. 3.2\)](#) for a complete discussion of the mislabeling.

By specifying `param(1)`, we obtain parameterization 1 for this model.

```

. pkcross outcome, param(1)

Sequence variable = sequence
Period variable = period
Treatment variable = treat
Carryover variable = carry
ID variable = id

```

Analysis of variance (ANOVA) for a 2x2 crossover study

Source of variation	Partial SS	df	MS	F	Prob > F
Treatment effect	301.04	1	301.04	0.67	0.4189
Period effect	255.62	1	255.62	0.57	0.4561
Carryover effect	276.00	1	276.00	0.61	0.4388
Residuals	19890.92	44	452.07		
Total	20265.68	47			

Omnibus measure of separability of treatment and carryover = 29.2893%

◀

► Example 2

Consider the case of a two-treatment, four-sequence, two-period crossover design. This design is commonly referred to as Balaam's design ([Balaam 1968](#)). [Ratkowsky, Evans, and Alldredge \(1993, 140\)](#) published the following data from an amantadine trial, originally published by [Taka and Armitage \(1983\)](#):

```
. use https://www.stata-press.com/data/r18/balaam, clear
. list, sep(0)
```

	id	seq	period1	period2	period3
1.	1	-ab	9	8.75	8.75
2.	2	-ab	12	10.5	9.75
3.	3	-ab	17	15	18.5
4.	4	-ab	21	21	21.5
5.	1	-ba	23	22	18
6.	2	-ba	15	15	13
7.	3	-ba	13	14	13.75
8.	4	-ba	24	22.75	21.5
9.	5	-ba	18	17.75	16.75
10.	1	-aa	14	12.5	14
11.	2	-aa	27	24.25	22.5
12.	3	-aa	19	17.25	16.25
13.	4	-aa	30	28.25	29.75
14.	1	-bb	21	20	19.51
15.	2	-bb	11	10.5	10
16.	3	-bb	20	19.5	20.75
17.	4	-bb	25	22.5	23.5

The sequence identifier must be a string with 0s to indicate washout or baseline periods, or a number. If the sequence identifier is numeric, the `order()` option must be specified with `pkshape`. If the sequence identifier is a string, `pkshape` will use the string values to create sequence, period, and treatment variables. In this example, the dash is used to indicate a baseline period, which is an invalid code for this purpose. Therefore, we use the `subinstr()` function to replace each dash with a 0. After doing so, we can use `pkshape` to format the data in a way that can be used with `pkcross`. We leave most `pkcross` options at their defaults, but we specify the `sequential` option to calculate sequential sums of squares instead of the default partial sums of squares.

```
. replace seq = subinstr(seq, "-", "0", .)
(17 real changes made)
. pkshape id seq period1 period2 period3
. pkcross outcome, sequential
```

```
Sequence variable = sequence
Period variable = period
Treatment variable = treat
Carryover variable = carry
ID variable = id
```

Analysis of variance (ANOVA) for a crossover study

Source of variation	SS	df	MS	F	Prob > F
<hr/>					
Intersubjects					
Sequence effect	285.82	3	95.27	1.01	0.4180
Residuals	1221.49	13	93.96	59.96	0.0000
<hr/>					
Intrasubjects					
Period effect	15.13	2	7.56	6.34	0.0048
Treatment effect	8.48	1	8.48	8.86	0.0056
Carryover effect	0.11	1	0.11	0.12	0.7366
Residuals	29.56	30	0.99		
<hr/>					
Total	1560.59	50			

Omnibus measure of separability of treatment and carryover = 64.6447%

▷ Example 3

For this example, we return to `pkdata.dta` from [example 2](#) of [\[R\] pk](#) and use `pkcollapse` and `pkshape` on the data as discussed in that example.

```
. use https://www.stata-press.com/data/r18/pkdata, clear
(Fictional drug concentration data)
. pkcollapse time conc1 conc2, id(id) keep(seq) stat(auc)
.....
. pkshape id seq auc*, order(RT TR)
```

After sorting the data with `sort`, our data appear as follows:

```
. sort period id
. list, sep(8)
```

	id	sequence	outcome	treat	carry	period
1.	1	RT	150.9643	R	0	1
2.	2	RT	146.7606	R	0	1
3.	3	RT	160.6548	R	0	1
4.	4	RT	157.8622	R	0	1
5.	5	RT	133.6957	R	0	1
6.	6	RT	160.639	R	0	1
7.	7	RT	131.2604	R	0	1
8.	8	RT	168.5186	R	0	1
9.	9	TR	137.0627	T	0	1
10.	10	TR	153.4038	T	0	1
11.	11	TR	163.4593	T	0	1
12.	12	TR	146.0462	T	0	1
13.	13	TR	158.1457	T	0	1
14.	14	TR	147.1977	T	0	1
15.	15	TR	164.9988	T	0	1
16.	16	TR	145.3823	T	0	1
17.	1	RT	218.5551	T	R	2
18.	2	RT	133.3201	T	R	2
19.	3	RT	126.0635	T	R	2
20.	4	RT	96.17461	T	R	2
21.	5	RT	188.9038	T	R	2
22.	6	RT	223.6922	T	R	2
23.	7	RT	104.0139	T	R	2
24.	8	RT	237.8962	T	R	2
25.	9	TR	139.7382	R	T	2
26.	10	TR	202.3942	R	T	2
27.	11	TR	136.7848	R	T	2
28.	12	TR	104.5191	R	T	2
29.	13	TR	165.8654	R	T	2
30.	14	TR	139.235	R	T	2
31.	15	TR	166.2391	R	T	2
32.	16	TR	158.5146	R	T	2

We now fit an ANOVA model using pkcross:

```
. pkcross outcome
```

```
Sequence variable = sequence
Period variable = period
Treatment variable = treat
Carryover variable = carry
ID variable = id
```

Analysis of variance (ANOVA) for a 2x2 crossover study

Source of variation	Partial SS	df	MS	F	Prob > F
Intersubjects					
Sequence effect	378.04	1	378.04	0.29	0.5961
Residuals	17991.26	14	1285.09	1.40	0.2691
Intrasubjects					
Treatment effect	455.04	1	455.04	0.50	0.4931
Period effect	419.47	1	419.47	0.46	0.5102
Residuals	12860.78	14	918.63		
Total	32104.59	31			

Omnibus measure of separability of treatment and carryover = 29.2893%

◀

▶ Example 4

Consider the case of a six-treatment crossover trial in which the squares are not variance balanced. The following dataset is from a partially balanced crossover trial published by [Patterson and Lucas \(1962\)](#) and reproduced in [Ratkowsky, Evans, and Alldredge \(1993, 231\)](#):

```
. use https://www.stata-press.com/data/r18/nobalance, clear
. list, sep(4)
```

	cow	seq	period1	period2	period3	period4	block
1.	1	adbe	38.7	37.4	34.3	31.3	1
2.	2	baed	48.9	46.9	42	39.6	1
3.	3	ebda	34.6	32.3	28.5	27.1	1
4.	4	deab	35.2	33.5	28.4	25.1	1
5.	1	dafc	32.9	33.1	27.5	25.1	2
6.	2	fdca	30.4	29.5	26.7	23.1	2
7.	3	cfad	30.8	29.3	26.4	23.2	2
8.	4	acdf	25.7	26.1	23.4	18.7	2
9.	1	efbc	25.4	26	23.9	19.9	3
10.	2	becf	21.8	23.9	21.7	17.6	3
11.	3	fceb	21.4	22	19.4	16.6	3
12.	4	cbfe	22.8	21	18.6	16.1	3

When there is no variance balance in the design, a square or blocking variable is needed to indicate in which treatment cell a sequence was observed, but the mechanical steps are the same.

```
. pkshape cow seq period1 period2 period3 period4
. pkcross outcome, model(block cow|block period|block treat carry) sequential
      Number of obs =      48      R-squared      = 0.9965
      Root MSE      =  .740408      Adj R-squared = 0.9903
```

Source	Seq. SS	df	MS	F	Prob > F
Model	2650.1331	30	88.3377701	161.14	0.0000
block	1607.01128	2	803.505642	1465.71	0.0000
cow block	628.706274	9	69.8562527	127.43	0.0000
period block	408.031253	9	45.3368059	82.70	0.0000
treat	2.50000057	5	.500000114	0.91	0.4964
carry	3.88428906	5	.776857812	1.42	0.2680
Residual	9.31945887	17	.548203463		
Total	2659.45256	47	56.584097		

When the model statement is used and the omnibus measure of separability is desired, specify the variables in the `treatment()`, `carryover()`, and `sequence()` options to `pkcross`. ◀

Methods and formulas

`pkcross` uses ANOVA to fit models for crossover experiments; see [R] [anova](#).

The omnibus measure of separability is

$$S = 100(1 - V)\%$$

where V is Cramér's V and is defined as

$$V = \left\{ \frac{\frac{\chi^2}{N}}{\min(r-1, c-1)} \right\}^{\frac{1}{2}}$$

N is the sample size and χ^2 is calculated as

$$\chi^2 = \sum_i \sum_j \left\{ \frac{(O_{ij} - E_{ij})^2}{E_{ij}} \right\}$$

where O and E are the observed and expected counts in an $r \times c$ table of the number of times each treatment is followed by the other treatments.

References

- Balaam, L. N. 1968. A two-period design with t^2 experimental units. *Biometrics* 24: 61–73.
<https://doi.org/10.2307/2528460>.
- Chow, S.-C., and J.-P. Liu. 2009. *Design and Analysis of Bioavailability and Bioequivalence Studies*. 3rd ed. Boca Raton, FL: Chapman and Hall/CRC.

- Kutner, M. H., C. J. Nachtsheim, J. Neter, and W. Li. 2005. *Applied Linear Statistical Models*. 5th ed. New York: McGraw–Hill/Irwin.
- Patterson, H. D., and H. L. Lucas. 1962. Change-over designs. Technical Bulletin 147, North Carolina Agricultural Experiment Station and the USDA.
- Ratkowsky, D. A., M. A. Evans, and J. R. Alldredge. 1993. *Cross-over Experiments: Design, Analysis, and Application*. New York: Dekker.
- Taka, M. T., and P. Armitage. 1983. Autoregressive models in clinical trials. *Communications in Statistics—Theory and Methods* 12: 865–876. <https://doi.org/10.1080/03610928308828502>.

Also see

[R] **pk** — Pharmacokinetic (biopharmaceutical) data

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the [FAQ on citing Stata documentation](#).